

# Aus David wird Goliath

512 KByte RAM für Schneider CPC

Hubert Schröder / Gabor Herr

Wer träumt nicht von einem riesenhaften Speicher für seinen Computer. Ein halbes Megabyte würde bei einem 8-Bit-Computer wie dem Schneider CPC schon das Adjektiv 'riesig' verdienen. Wenn Sie nicht davor zurückschrecken, Hand an die RAM-Chips zu legen, zeigt Ihnen der folgende Beitrag, wie Sie aus Ihrem CPC 6128 einen 'CPC 6512' machen können. Halt, 664- und 464-Besitzer, hören Sie noch nicht auf zu lesen – in weiteren Beiträgen kommen auch Sie an die Reihe.

Bei einer RAM-Disk mit 384 KByte Kapazität bestünde nur noch das Problem, wie diese zu füllen wäre; und durch größere Sektoren- und Druckerpuffer würden zum Beispiel die obligatorischen Kaffeepausen beim Laden, Kompilieren oder

Drucken von Programmen drastisch verkürzt.

Da CP/M Plus, im Gegensatz zu CP/M 2.2, von Hause aus die Verwaltung eines so großen Speicherbereichs unterstützt, ist die Aufrüstung ohne allzu große

Änderung der bereits vorhandenen Soft- und Hardware realisierbar. Es brauchen nur die eingebauten 64-KBit-Chips durch die neuen 256-KBit-DRAMs ausgetauscht zu werden. Bei den heutigen Speicherpreisen ist diese Lösung sogar noch billiger als der Kauf eines Floppy-Laufwerkes, das auch noch viel langsamer ist als die nun verfügbare RAM-Disk.

Der CPC 6128 ist als abgeschlossenes und nicht mehr erweiterbares System entwickelt worden; daraus resultieren bei Verwendung größerer Chips zwei Probleme: die Generierung eines 8-Bit-Refresh und die Verwaltung der neu hinzugekommenen Adreßleitungen A16 und A17.

## Erfrischungen

Was ist überhaupt ein Refresh? Bei dynamischen RAMs werden die Informationen durch aufbeziehungsweise Entladung eines Kondensators gespeichert. Da sich Kondensatoren nach einer bestimmten Zeit selbständig entladen, muß die Ladung zyklisch aufgefrischt, 'refreshed', werden. Dies geschieht durch Zugriff auf die einzelnen Speicherzellen, die in einer Matrix aus Zeilen und Spalten organisiert sind.

Deshalb erfolgt der Zugriff 'multiplexed', das heißt, es wird zuerst die Zeilenadresse (Row Address) an die Adreßeingänge gelegt. Nach einem RAS-Signal (Row Address Strobe) wird die so angesprochene Zeile in ein internes Register kopiert. Danach wird auf die gleiche Art die Spaltenadresse (Column Address) übergeben, die ein Bit aus diesem Register selektiert und zum Datenausgang überträgt. Der eigentliche Refresh findet bereits nach dem RAS-Signal statt, indem die durch das Register verstärkten Ladungen wieder zurück in die Speicherzeile übertragen werden.

Weil der CPU-Zugriff auf die einzelnen Zeilen zeitlich zufällig erfolgt, wird durch einen Refresh-Zähler eine zyklische Adressierung der einzelnen Zeilen gewährleistet. Dies kann aber nur dann stattfinden, wenn die CPU gerade nicht auf die DRAMs zugreift. (Beim Z80 synchronisiert das REFRESH-Signal die Zugriffe.)

Die 64-KBit-DRAMs sind zu 2<sup>7</sup> Zeilen und 2<sup>9</sup> Spalten organi-

siert, so daß ein 7-Bit-Refresh ausreicht, der von der Z80-CPU generiert wird. Dagegen benötigen die 256-KBit-DRAMs einen 8-Bit-Refresh, da die Matrix 2<sup>8</sup> Zeilen hat.

Beim CPC jedoch wird der Refresh-Mechanismus der Z80-CPU nicht genutzt. Diese Aufgabe hat der Video-Controller übernommen, der sich den Speicher mit der CPU 'teilt', das heißt, beide greifen – synchronisiert durch das CPU-Signal – abwechselnd auf den Speicher zu. Weil der Video-Controller die ausgelesenen Daten für den Bildschirmaufbau verwendet, werden nacheinander alle Adressen des Bildschirmspeichers angesprochen. Dies würde einem 14-Bit-Refresh entsprechen, bei dem natürlich auch die Bedingungen eines 8-Bit-Refresh erfüllt sind. Diese technische Besonderheit des Schneiders ist überhaupt erst die Voraussetzung für den problemlosen Austausch der Speicher-ICs gegen größere.

## Hausverwalter PAL

Die Z80-CPU besitzt bekanntlich nur einen Adreßraum von 64 KByte, weshalb ein größerer Speicher in Bänke aufgeteilt werden muß (siehe 'Des Schneiders Kern', Teil 1, c't 8/85). Die Amstrad-Entwickler lösten dieses Problem mit einem PAL 16L8. Dieser Baustein ist logisch ein internes Register des Gate-Array und wird über die Port-Adresse 7FXXh bei gesetzten Bits 6 und 7 angesprochen. Die unteren drei Datenbits dieses Banking-Registers selektieren die Speicherkonfigurationen 0 bis 7 (siehe Tabelle der Speicherkonfiguration). Das PAL hat folgende Aufgaben zu erfüllen:

- Speichern der zum 'Bank-Switching' erforderlichen drei Datenbits
- 'Umleitung' der Signale A14, A15 und CAS an die physikalischen 16-KByte-Blöcke, entsprechend der eingestellten Speicherkonfiguration
- Selektierung der Bank 0 bei inaktivem CPU (damit der 6845 auf den Videospeicher zugreifen kann)

## Auf- oder Abrüsten?

Bei näherer Betrachtung des Schaltplanes fällt ins Auge, daß die Amstrad-Entwickler eine geniale Idee in die Tat umgesetzt

	Nr. der Konfiguration								
	0	1	2	3	4	5	6	7	8...31
BLOCK 3	3	7	7	7	3	3	3	3	7
2	2	2	6	2	2	2	2	2	2
1	1	1	5	3	4	5	6	7	8-31
0	0	0	4	0	0	0	0	0	0
BASIC-Einschaltkonfiguration	CP/M		Sektor-Buffer	BASIC-Konfiguration, mit den Blöcken 4 bis 7				Erweiterter Adreßbereich durch das neue PAL	

Die neue PAL-Logik blendet in der üblichen CP/M-Plus-Konfiguration (2) die neuen Blöcke nacheinander ein. Diese Konfigurationen (8 bis 31) können nur durch eigene Routinen angesprochen werden. Gelb gekennzeichnet ist das Common-RAM unter CP/M.

haben. Der 6128 läßt sich durch Entfernen des PALs und Setzen der Drahtbrücken LK 5, 6 und 8 zu einem CPC 664 umbauen. Es liegt die gleiche Situation vor wie bei gewissen Mittelstreckenraketen, die auch problemlos zu einer kleineren Version umgebaut werden können.

Die Umbaurichtung läßt sich logischerweise genauso einfach umdrehen.

Im Banking-Register sind insgesamt 6 Bit frei verfügbar; es ließen sich also  $2^6 \cdot 16 \text{ KB} = 1 \text{ MB}$  adressieren. Die Erweiterung beschränkt sich aber auf 512 KByte, weil auf der Platine des 6128 nur 16 DRAMs Platz haben. Diese Ausbaustufe ist für die meisten Anwendungen ausreichend.

### LötKolben anwärmen

Bevor es losgeht, noch einige Warnungen: Wer noch Garantie auf seinem Rechner hat, verliert sie natürlich nach einem

Vier zusätzliche Anschlüsse erfordert die Erweiterung. Wohin sie gehören, verrät der Schaltplan.

solchen Eingriff. Wer dennoch diese 'Operation am offenen Herzen' durchführen will, sollte dies mit äußerster Geduld und Vorsicht tun. Dadurch spart man sich in der Regel die mühsame und nervenaufreibende Fehlersuche nach kalten Lötstellen, Leiterbahnunterbrechungen und anderen Katastrophen. Die Verwendung von geeignetem Werkzeug erhöht die Überlebenschance des Rechners erheblich.

Sind die Gehäuseschrauben entfernt, läßt sich die Haube mit der Tastatur abnehmen. Da die Tastatur noch über Kabel mit der Mutterplatine verbunden ist, läßt sie sich vorerst nur zur Seite klappen. Nach dem Entfernen des Abschirmbleches können Sie nun alle steckbaren Verbindungen auf der Mutterplatine lösen. Beim Stecker für das interne Floppy-Drive sollte man dies allerdings nicht versuchen: er ist mit der Platine verlötet. Alle Stecker sind mechanisch und farblich kodiert, somit ist ein Vertauschen der Steckverbinder beim Zusammenbau nicht möglich. Um bei eventuell auftretenden Störungen die Fehlersuche auf nur eine Bank eingrenzen zu können, wird zuerst nur eine Bank, und zwar Bank 0 (ICs 127 bis 134), ausgelötet.

Kann man auf die ausgelöteten ICs verzichten, so kneift man

die 'Beinchen' möglichst nahe am IC-Gehäuse ab. Diese Methode ist einfacher und daher vorzuziehen. Jetzt kann man nämlich mit dem LötKolben jedes 'Beinchen' einzeln erhitzen und mit der Spitzzange oder einer Pinzette herausziehen. Danach werden mit der Entlösaugpumpe von der Unterseite der Mutterplatine her die Löt-Augen freigesaugt und mit der Entlösauglitze Lötzinnreste beseitigt.

Vor der optischen Zwischenkontrolle sollte man die bearbeiteten Flächen der Platine mit einem Lösungsmittel vom Kolophonium reinigen, um eventuell vorhandene Kurzschlüsse oder Unterbrechungen besser erkennen zu können.

Hat man etwaige Fehler beseitigt, können die IC-Sockel eingesetzt und verlötet werden. Hier stolpert man wieder über die unkonventionelle Denkweise der Entwickler dieses Rechners: ICs werden in der Regel mit leicht nach außen gebogenen 'Beinchen' geliefert. Anstatt aber nun beim Bestücken die ICs zurechtzubie-


gen, vergrößerten sie einfach den Abstand des Lötspiegels. Übliche IC-Sockel passen nun nicht mehr, deren Stege müssen vor dem Einbau mit der Zange durchtrennt werden.


Auf der Rückseite der Platine werden jetzt alle Pins 1 miteinander verbunden und auf Massepotential gelegt. Dazu verwendet man am besten einen isolierten Draht, den man in Abständen von ungefähr 1,5 cm abisoliert hat.

### Fehlersuche

Haben Sie es bis hierhin geschafft, dann schließen Sie die Mutterplatine wieder an allen lebenswichtigen Kabeln an und schalten den Rechner ein. Meldet sich der Computer wie gewohnt, dann haben Sie die erste Schlacht bereits gewonnen. Falls Sie den Rechner ohnehin nur bis 320 KB aufrüsten wollen, können Sie gleich mit dem Einbau der Zusatzplatine beginnen.

Sollte sich allerdings das Schneiderlein recht eigensinnig verhalten, beginnt die Fehlersuche. Weigert er sich lediglich, ihre Befehle entgegenzunehmen, so haben Sie vielleicht vergessen, die Tastatur einzustecken. Wenn ein Oszilloskop zur Verfügung steht, sollte man zuerst die Form der Signale an den Daten- und Adreßleitungen der DRAMs abgreifen. Treten hierbei ungewohnte Signalformen auf (zum Beispiel Pegel auf 2-V-Potential), so ist zunächst das betreffende IC aus der Fas-





Ein (Hilfs-)Schaltplan zeigt die logische Zuordnung von RAM, Banking-Logik und CPU.

mit BANKMAN installiert wird, einem Programm, das auf jeder Betriebssystemdiskette vorhanden ist. 'Iscreencopy,1,x' kopiert den Inhalt des Bildschirmspeichers in einen anderen der vier möglichen 16-KByte-Blöcke (x); 'Iscreencopy,x,1' holt das Bild wieder zurück.

Zum Test kopiert ein kleines BASIC-Programm zuerst einen hellen, dann einen dunklen Bildschirm in den zu prüfenden Block. Zur Sicherheit sollte man die Reihenfolge von hell/dunkel einmal umkehren, denn ob das Speicher-IC eine unterbrochene Leitung als High oder als Low interpretiert, hängt vom Zufall ab.

Bei einer defekten Bank zeigt das zurückgeholte Bild falsch gesetzte Pixel (siehe Fotos). Unterbrochene Datenleitungen verursachen vertikale Streifen auf dem Bildschirm. Der rechte Rand des Cursors entspricht D0. Fehler, die sich nicht auf den ganzen Bildschirm auswirken, werden von Adreßleitungen ausgelöst. Solche Ausschnittfehler, die auch von vertikalen Störungen durchsetzt sind, deuten auf eine nur teilweise Unterbrechung der Adreßleitung hin.

Nachdem man schließlich seinen Computer mit dem Lötkolben überzeugt hat, seine 'normale Arbeit' wieder aufzunehmen, wird die kleine Zusatzplatine in Angriff genommen und


sung zu nehmen. Sieht das Signal wieder 'normal' aus, müßte sich der Fehler durch Austauschen des IC beheben lassen.

Ansonsten muß man dem Fehler mit dem Ohmmeter zu Leibe rücken und die betreffende Leiterbahn auf Unterbrechung beziehungsweise auf Kurzschluß mit allen benachbarten Leiterbahnen überprüfen. Hierzu dient der Schaltplanauszug, der alle Verbindungen der DRAM-Bank zum Rest des Computers auflistet.

Fehlt ein Oszilloskop in der Ausrüstung, so muß man sich auf die Fehlersuche mit dem Ohmmeter beschränken. Dies ist jedoch kein so großer Nachteil, da beim Booten des Rechners die CPU auf eine defekte RAM-Bank zugegriffen hat und sich deshalb in einem unbekanntem Zustand befindet. Die logische Aussage der Signale, die man vor allem auch mit einem mehrstrahligen Oszilloskop erkennen kann, ist deshalb wertlos.

Nach einwandfreier Funktion des Rechners führen Sie die glei-


che Aktion noch einmal für die zweite Bank durch, die Fehlersuche eingeschlossen. Ein Fehler (vielleicht Kurzschluß) in der zweiten wirkt sich auch auf die Funktion der ersten Bank aus, weil die beiden Speicherbänke (bis auf das CAS-Signal) vollkommen parallel verdrahtet sind. Ab und zu kommt es auch vor, daß ein defektes IC eine oder mehrere Leitungen auf einen festen Pegel 'zieht'.



Nur mit einem neuen PAL ist es nicht getan. Diese Platine nimmt das neue PAL auf und wird mit Stiftleisten in den Sockel des alten eingesteckt.

### Zweite Runde

Alle Fehler beseitigt? dann wird es jetzt Zeit, CP/M zu booten. Sollte dabei der CPC einen Restart durchführen, arbeitet nur die zweite Bank noch nicht korrekt, der Rest des Rechners ist jedoch voll funktionstüchtig. Daher läßt sich der Fehler softwaremäßig lokalisieren. Zu diesem Zweck ist der RSX-Befehl 'Iscreencopy' ganz nützlich, der



### Stückliste

ICs	
IC 1	74LS38
IC 2	PAL 16L8
IC 3	74LS273

#### Widerstände


R 1	2k2
R 2	470R
R 3	47R

#### Sonstiges

Silberdraht, Durchmesser ca. 0,6 mm

#### Werkzeuge:

- Spitzzange (notfalls Pinzette)
- Seitenschneider
- Lötkolben (ideal 25-30 Watt, feine Lötspitze)
- Entlötpumpe und/oder Entlötlitze
- Lösungsmittel (Spiritus, Aceton)
- Oszilloskop (nicht unbedingt notwendig)
- üblicher Krimskrams wie Schraubendreher ...



anhand des Bestückungsplanes zusammengebaut. Dabei sollte eine möglichst geringe Bauhöhe angestrebt werden, damit sich der Gehäusedeckel des CPC schließen läßt. Der Verzicht auf IC-Sockel bringt auch schon einige Millimeter.

Den Rest kann man einsparen, indem man den Steckkontakt für den PAL-Sockel aus kurzen Drahtenden fertigt. Anstelle des alten PALs wird nun die Zusatzplatine mit den Drahtstiften in

den freigewordenen IC-Sockel gesteckt. Die vier zusätzlich benötigten Verbindungen werden anhand der Anschluß-tabelle und den Bezeichnungen auf dem Bestückungsplan hergestellt.

Nach erneutem Anschalten des Rechners müßte wie gewohnt die Einschaltmeldung erscheinen. Falls nicht, sind alle Verbindungen, insbesondere der Steckkontakt zum PAL-Sockel, zu überprüfen.

Wer es bis hierhin geschafft hat, ist zwar ein halber Speicher-Millionär, kann jedoch mit den vielen kleinen grauen Zellen kaum etwas anfangen. Damit die Wartezeit bis zum Erscheinen des RAM-Disk-Treibers in der nächsten c't nicht zu lang wird, ist ein kleines Beispielprogramm abgedruckt, mit dem sich eine Folge von Bildern als Zeichentrickfilm darstellen läßt. Es mag auch als Anregung für eigene Programmwürfe dienen. (bb)

**Von solchen Bildern bleiben Sie hoffentlich verschont. Sie zeigen spezielle Fehler beim RAM-Austausch. Bei uns lief alles glatt, weshalb wir die Fehler zur Anschauung simulieren mußten.**

```

D7D6 D0 D3 D4 D1 D2 NCAS A15 A14 GND
CPU A15S AMUX MUX LCLK CAS1 CAS0 IOWR A14S VCC

IF (VCC) /LCLK= D7D6 * /A15 * /IOWR
IF (VCC) /CAS0= /NCAS * /D4 +
                /NCAS * A15 +
                /NCAS * /A14 +
                /NCAS * CPU
IF (VCC) /CAS1= /NCAS * D4 * /A15 * A14 * /CPU
IF (VCC) /A14S= /A14 +
                /D0 * D2 * /A15 +
                /D0 * D3 * /A15 +
                /D0 * D4 * /A15
IF (VCC) /A15S= /A14 * /A15 +

                /D1 * /A15 +
                /D4 * /D3 * /D2 * /D0 * /A15 +
                /D4 * /D3 * /D2 * /D1 * /A15
IF (VCC) /AMUX= /D0 * D1 * /D2 * /D3 * /D4 * /CPU * /MUX +
                A15 * A14 * /D2 * D0 * /CPU * /MUX +
                A15 * A14 * /D2 * D1 * /CPU * /MUX +
                A15 * A14 * D3 * /CPU * /MUX +
                A15 * A14 * D4 * /CPU * /MUX +
                /A15 * A14 * D2 * /CPU * /MUX +
                /A15 * A14 * D3 * /CPU * MUX
IF (GND) /MUX = /MUX
IF (GND) /IOWR= /IOWR
    
```

**Kein neues PAL ohne neue Logik; es wollen nun 512 KByte verwaltet werden.**

```

.z80
cseg

MaxBnk equ 29 ;Speicherausbau: 320KB --> 17,
; 512KB --> 29
Blklen equ 1000h ;Puffergroesse
blocks aset 4000h/Blklen

LogExt equ 0bcd1h ;KL LOG EXT, RSX einhaengen
ClrWin equ 0bb6ch ;TXT CLEAR WINDOW
; Anfangsadresse des Bildschirmspeichers setzen:
SetOffs equ 0bc05h ;SCR SET OFFSET
SetBase equ 0bc08h ;SCR SET BASE
SetPos equ 0bd55h ;SCR SET POSITION, 'Hintergrund'-
; Bildschirm aktivieren

.phase 8000h ;Adresse des RSX

initRSX:ld bc,RSXhead ;RSX initialisieren
ld hl,RSXchn
jp logext

RSXhead:dw Name ;AMSTRAD RSX Header:
jp scopy ;Routineneinspruege:
jp setscr
jp setbck
Name: dc "SCREENCOPY";Namen der Befehle
dc "SETSCREEN"
dc "SETBACK"
db 0

RSXchn: ds 4 ;Platz fuer die Kette

;SCREENCOPY,x,y
;-----
;Kopiert den Bildschirminhalt von x nach y, wie in BANKMAN,
;es sind aber die Parameter 6 bis 29 moeglich. Um den Kopier-
;vorgang im Hintergrund ablaufen zu lassen, wurde ein zweiter,
;"unsichtbarer" Bildschirmspeicher eingerichtet, der mit
;Block 0 angesprochen wird.

scopy: cp 2 ;zwei Parameter?
ret nz ;wenn nicht - Abbruch
    
```

```

call getpar ;2. Parameter (Ziel) holen
ld e,a ;und in e ablegen
inc ix
inc ix
call getpar ;dann 1. Parameter (Quelle)
ld d,a ; nach d

di ;Interrupts sperren
ld b,blocks ;Anzahl der Bloecke im b
ld hl,4000h ;ab Adresse 4000h

scopy2: push bc ;Blockzaehler retten
ld a,d ;Quellbank einschalten
call selbnk
ld a,e

push de ;Banknummer retten
push hl ;Quelladr. retten
ld de,Buffer ;1 KB ins Buffer
ld bc,Blklen ;kopieren
ldir
call selbnk ;Zielbank einschalten
pop de ;Zieladr. holen
ld hl,Buffer ; und Bufferinhalt
ld bc,Blklen ; dorthin kopieren
ldir
ex de,hl ;naechste Quelladr. in hl
pop de ;Banknr. wieder in de

pop bc ;Blockzahler rest.
djnz scopy2 ;fuer alle Bloecke wiederholen

call selbnk0 ;Systembank ein
ei ;Interrupts frei
ret ;und fertig...

;SETSCREEN,x
;-----
;Definiert Block x als Bildschirmspeicher, d.h. man kann
;zwischen den beiden Bildschirmspeichern umschalten.
;(Nur Block 0 oder 1 sind erlaubt.)

setscr: cp 1 ;nur ein Parameter erlaubt
ret nz

call getpar2 ;Parameter holen
push af ;Base retten

```

```

call setoffs ;Offset setzen
pop af ;Base laden
jp setbase ;und setzen.

;SETBACK,s
;-----
;Der mit s bezeichnete Bildschirmspeicher wird als "Hinter-
;grund"-Bildschirm definiert, d.h. die folgenden Text- und
;Grafikbefehle wirken auf diesen Bildschirm. Dargestellt
;wird jedoch der Inhalt des mit SETSCREEN selektierten
;Blocks.

setbck: cp 1 ;nur ein Parameter
ret nz

call getpar2 ;holen
call setpos ;und 'Hintergrund'-Screen setzen
jp clrwin

getpar: ld a,(ix) ;Wert kommt in den Akku
pop hl ;Ruecksprung vom Stack, ret=exit
cp MaxBnk+1 ;testen ob 0 >= a <= MaxBnk
ret nc
or a
ret m
push hl ;O.K., Ruecksprung wieder
ret nz ; auf den Stack
ld a,-2 ;wenn 0, dann a=-2 (<=> Bank 0)
ret

getpar2: ld a,(ix) ;Wert in den Akku
ld hl,0 ;Offset immer 0
or a ;Akku=0?
jr z,scr0 ;ja - dann weiter bei scr0,
ld a,0c0h ;sonst Bildschirm ab C000h
ret ;und fertig...

scr0: ld a,40h ;Bildschirm ab 4000h
ret

selbnk0: ld a,-2 ;BASIC-Bank selektieren

selbnk: add a,0c2h ;Selektiert Bank a
push bc ;bc retten
ld b,7fh ;b=Portadr. des Banking-Registers
out (c),a ;Banknr. hineinschreiben
pop bc ;bc restaurieren
ret

Buffer: ;hier beginnt der Blockpuffer

.dephase

end

```

```

10 'ScreenMan Loader
20 MEMORY &3FFF
30 LOAD "scrman.bin",&8000
40 CALL &8000
50 NEW

```

**Der Lader für das .BIN-File.**

Nach dem Umbau möchte man die neuen Bänke auch ausprobieren. Diese RSX-Erweiterung arbeitet ähnlich wie BANKMAN, kann aber jetzt mit sechs Bänken mehr operieren.

```

10 'demo
20 '
30 DEG: 0SETSCREEN,1: radius=50
40 FOR k=0 TO 28
50 CLS: GOSUB 2000
60 posx=320: posy=300-(k-14)*2*200/196: GOSUB 1000
70 0SCREENCOPY,1,k+2
80 NEXT
100 GOSUB 500
500 pict=0
510 screen=pict MOD 2: 0SETBACK,screen
520 0SCREENCOPY,pict+2,screen: 0SETSCREEN,screen
530 pict=pict+1: IF pict=28 THEN pict=0
540 GOTO 510
980 END
990 '
1000 'kugel
1010 hoehe=radius:
1020 FOR j=10*(k MOD 2) TO 90 STEP 20
1030 breite= radius*SIN(j) : GOSUB 1500

```

```

1040 NEXT
1050 RETURN
1500 'kreis
1505 MOVE posx,posy+hoehe
1510 FOR i=0 TO 360 STEP 10 :
1520 DRAW posx+breite*SIN(i),posy+hoehe*COS(i)
1530 NEXT
1550 RETURN
2000 'Hintergrund
2005 FOR i=10*(k MOD 4) TO 400 STEP 40
2010 MOVE 0.5*i,0.25*i: DRAW 400+0.5*i,0.25*i
2020 NEXT
2030 FOR i=0 TO 400 STEP 40
2040 MOVE i,0 : DRAW i+0.5*400,0.25*400
2060 NEXT
2100 RETURN

```

Mit diesem Beispielprogramm können sie sich die Zeit vertreiben bis zur nächsten c't.

